

Microsoft Excel: Lookup Functions

PC Class

Handouts available at glensidepld.org/technology

VLOOKUP

One of the most common tasks in Excel is the process of looking up specific values within a data set. This is commonly done using a **Function**. One of the most frequently used is **VLOOKUP**, which searches for a value in the leftmost column of your data range, finds a match, and returns a value in the same row of another column in the same data range.

VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

Description

What value are you searching for?

This is the lookup value. Excel will look for a match to this value in the leftmost column of your lookup table.

Where do you want to search?

This is the lookup table. If you plan to copy your VLOOKUP formula, you may want to use absolute references to "lock" the range.

Which column contains the search result?

This value will appear in the cell with the VLOOKUP formula. Count over from the first column to figure out what this number should be, starting with 1.

Should the lookup value be an exact match (FALSE or 0) or is an approximate match (TRUE or 1) okay if an exact match doesn't exist?

For TRUE, sort the leftmost column in ascending order for correct results.

Here we have a simple data range. It's a matrix consisting of rows that represent regions of the country and columns that show Income, Expenses, and Cash Flow for those regions.

	A	B	C	D	E
1					
2		Region	Income	Expenses	Cash Flow
3		Northeast	\$ 2,000,000.00	\$ 1,500,000.00	\$ 500,000.00
4		Mid-Atlantic	\$ 1,740,000.00	\$ 1,320,000.00	\$ 420,000.00
5		Southeast	\$ 3,400,000.00	\$ 1,800,000.00	\$ 1,600,000.00
6		Midwest	\$ 900,000.00	\$ 1,050,000.00	\$ (150,000.00)
7		Southwest	\$ 1,150,000.00	\$ 1,100,000.00	\$ 50,000.00
8		Pacific	\$ 4,800,000.00	\$ 2,000,000.00	\$ 2,800,000.00

Using VLOOKUP, we can search for a region and return the matching Income, Expenses, or Cash Flow.

=VLOOKUP("Southeast", \$B\$3:\$E\$8, 3, False)

We search for the region in the leftmost column of our table and return the matching value from column 3. We are specifying FALSE so we get an exact match on the region. This returns the value \$1,800,000.00, the Expenses for the Southeast region.

A Better Way: INDEX/MATCH

VLOOKUP is useful and time-tested, but it has limitations. What if the value you need to match is not in the leftmost column? In this case, it's impossible to use VLOOKUP to search for the region. Also, if you happen to add a column in the middle of your data range, your VLOOKUP will be broken. An alternative uses two distinct functions, **INDEX** and **MATCH** to get the same result.

MATCH searches for a value in an array and returns the relative position (row) of that item.

The syntax for the **MATCH** function is:

MATCH(value, array, [match_type])

value: The value to search for in the array.

array: A range of cells that contains the value that you are searching for.

match_type: Optional. It's the type of match that the function will perform.

The possible values are:

- 1 (default) The MATCH function will find the largest value that is less than or equal to value. You should be sure to sort your array in ascending order.
- 0 The MATCH function will find the first value that is equal to value. The array can be sorted in any order. **We will be using 0 in our examples.**
- 1 The MATCH function will find the smallest value that is greater than or equal to value. You should be sure to sort your array in descending order.

Example: =MATCH(B10,B3:B8,0) searches for an exact match for "Southeast" in the list of regions and returns the value 3 for row 3.

The **INDEX** function returns a value in a table based on the intersection of a row and column position within that table. The first row in the table is row 1 and the first column in the table is column 1.

The syntax for the INDEX function is:

INDEX(table, row_number, [column_number])

table: A range of cells that contains the table of data.

row_number: The row position in the table where the value you want to lookup is located. This is the relative row position in the table and not the actual row number in the worksheet.

column_number: Optional. The column position in the table where the value you want to lookup is located. This is the relative column position in the table and not the actual column number in the worksheet.

Example: =INDEX(C3:E8,3,3) returns the value found at the intersection of Row 3 and Column 3, which is the Expenses for the Southeast region, \$1,800,000.

NOTE: There is another form of the INDEX function, which returns a cell **reference** rather than a value. We will not discuss this form in this handout.

The power of the two functions comes when MATCH is **nested** within INDEX, in this case allowing us to do a lookup of Income, Expenses, or Cash Flow using only one formula.

Example: Suppose we supply the region in a cell of our worksheet; in this case B10. We choose to display Income, Expenses, or Cash Flow in D10.

	A	B	C	D	E	F
1						
2		Region	Income	Expenses	Cash Flow	
3		Northeast	\$ 2,000,000.00	\$ 1,500,000.00	\$ 500,000.00	
4		Mid-Atlantic	\$ 1,740,000.00	\$ 1,320,000.00	\$ 420,000.00	
5		Southeast	\$ 3,400,000.00	\$ 1,800,000.00	\$ 1,600,000.00	
6		Midwest	\$ 900,000.00	\$ 1,050,000.00	\$ (150,000.00)	
7		Southwest	\$ 1,150,000.00	\$ 1,100,000.00	\$ 50,000.00	
8		Pacific	\$ 4,800,000.00	\$ 2,000,000.00	\$ 2,800,000.00	
9						
10	Region:	Southeast	Category:	Expenses	Amount:	\$ 1,800,000.00
11						

Our new formula looks like this:

=INDEX(C3:E8,MATCH(\$B\$10,B3:B8,0),MATCH(\$D\$10,C2:E2,0))

The first MATCH function, in **blue**, looks up region “Southeast” and determines it’s in row **3**. The second MATCH, in **green**, locates the category “Expenses” and returns column **3**. Both these are input to the INDEX function, which is now equivalent to **=INDEX(C3:E8,3,3)**.

Notice that there are only two **absolute references** used in these functions. We’re only locking the region input (\$B\$10) and the category input (\$D\$10). Everything else is **relative**. What this means is that if we insert a column, say, between Income and Expenses or a row between Southeast and Midwest, our INDEX/MATCH will continue to work.

That would cause big problems for a VLOOKUP.

Additionally, there are performance benefits to using INDEX/MATCH instead of VLOOKUP. That’s not a big deal for our tiny example but when you start doing calculations on multiple thousands of rows, it can make a difference.

NOTE: This is actually more advanced than a standard INDEX/MATCH. This is an **INDEX/MATCH/MATCH**, which does a **Matrix lookup** (both the row and column references come from a MATCH). To write this as an INDEX/MATCH that returns only “Expenses” we would have:

=INDEX(D3:D8,MATCH(B10,B3:B8,0))

Other Useful Lookup and Reference Functions

Name	Sample Formula	What It Does
OFFSET	=OFFSET(B2,1,0)	Returns a range that is a specified number of rows and columns from a reference cell or range.
HLOOKUP	=HLOOKUP("blue",A1:C5,2,0)	Looks for a value in the top row of a range and returns a value from the same column in a row you specify (Think VLOOKUP turned 90 degrees).
ROWS	=ROWS(A10:D18)	Returns the number of rows in a range of cells.
COLUMNS	=COLUMNS(A10:D18)	Returns the number of columns in a range of cells.
TRANSPOSE	=TRANSPOSE(A1:A25)	Converts a vertical range into a horizontal range and vice versa.

Many of these functions are used as inputs to, or in conjunction with, other functions. There are a number of useful lookup formulas that use these functions. INDEX/MATCH is one. Others include:

- VLOOKUP/MATCH
- VLOOKUP/HLOOKUP
- OFFSET/MATCH/MATCH
- INDEX/MATCH/MATCH (see previous example)

Room for Notes: